
10 CLS
20 KEY OFF
30 PRINT "PROGRAMA PARA LA SOLUCION DEL ALGORITMO DE DEMOUCRON"
40 PRINT"Y DIVIDIR EN NIVELES UN GRAFO PERT, CON EXPRESION DE LA"
50 PRINT "MATRIZ DE RESTRICCIONES PARA DISTRIBUCION EN PLANTA (MD1.BAS)"
60 REM    Autor> MARINO MARCOS CUERVO MAYO 2015
70 PRINT"COMIENZA CICLO DE PETICION DE DATOS"
80 PRINT
90 INPUT "Elementos de la matriz cuadrada=";NT
100 PRINT
110 REM ******************************
120 PRINT"SI EXISTE UN ARCO QUE UNE EL ELEMENTO PROBLEMA CON OTRO, SE MARCA 1"
130 PRINT "SI NO EXISTE, SE MARCA CON 0 O RETURN"
140 PRINT "SI EXISTE RESTRICCION POSITIVA DEL ELEMENTO PROBLEMA CON OTRO (CERCANIA) SE MARCA 2"
150 PRINT "SI EXISTE RESTRICION NEGATIVA (DISTANCIA) SE MARCA 9"
160 PRINT
170 INPUT "Para comenzar la entrada de Datos, debe tener delante el grafo PERT y pulsar RETURN";V
180 CLS
190 N=NT
200 DIM E(N,N)
210 DIM ES(N,N)
220 DIM R(N,N)
230 DIM V(N,N)
240 DIM EL(N)
250 DIM G(N,N)
260 DIM F(N)
270 DIM T(N,N)
280 DIM X(N,N)
290 DIM S(N,N)
300 REM *** COMIENZO ESCRITURA DE LA MATRIZ ASOCIADA AL GRAFO
310 REM *** Y DE LA MATRIZ DE RESTRICCIONES *****************
320 FOR C1=N TO 1 STEP -1
330 PRINT "Suceso";C1
340 C=N-1
350 FOR C2=C TO 1 STEP -1
360 PRINT "            existe arco entre";C1;"y";C2;" SI(1), NO(0)"
370 INPUT "                                                       ";E(C1,C2)
380 PRINT "            existe RESTRICCION"
390 INPUT "                                                       ";X(C1,C2)
400 NEXT C2
410 N=N-1
420 IF N=1 THEN GOTO 490
430 REM *** CUANDO TERMINA UN SUCESO, BORRA LA PANTALLA
440 REM *** Y PASA AL SIGUIENTE. **********************

450 INPUT "PARA SIGUIENTE SUCESO, PULSE RETURN";H
460 IF H=345 THEN STOP
470 CLS
480 NEXT C1
490 N=NT
500 CLS
510 REM SALIDA DE LA MATRIZ ASOCIADA AL GRAFO
520 PRINT " MATRIZ ASOCIADA AL GRAFO "
530 PRINT
540 REM *************************************
550 FOR C3=1 TO N
560 FOR C4=1 TO N
570 PRINT E(C4,C3);
580 NEXT C4
590 PRINT
600 NEXT C3
610 PRINT
620 PRINT "-----------------------------"
630 INPUT "PARA NIVELACION PULSE RETURN";G
640 IF G<>6789 THEN GOTO 710
650 REM COMIENZO DE LA NIVELACION
660 REM *************************
670 REM AMPLIAMOS LA MATRIZ ASOCIADA AL GRAFO PERT CON UN VECTOR SUMA DE
680 REM LAS FILAS DE LA MATRIZ
690 REM **********************
700 PRINT
710 B=1
720 PRINT "Nivel=";B
730 FOR F=1 TO N
740 FOR C=1 TO N
750  ES(1,F)=E(C,F)+ES(1,F)
760 V(1,F)=ES(1,F)
770 NEXT C
780 NEXT F
790 X=NT
800 PRINT "PT n§=";X
810 NA=B
820 B=B+1
830 D=0
840 N=NT
850 PRINT "--------------------
860 PRINT "Nivel=";B
870 PRINT
880 FOR F=1 TO N
890 IF B=2 THEN GOTO 920
900 V(B,F)=V(NA,F)-G((B-1),F)
910 IF B>2 THEN GOTO 930
920 V(B,F)=V(NA,F)-E(X,F)
930 IF V(B,F)<0 THEN V(B,F)=0
940 IF V(NA,F)=0 THEN GOTO 970
950 REM PRINT V(B,F)
960 IF V(B,F)=0 THEN GOSUB 1150
970 NEXT F
980 FOR D=1 TO M
990 FOR C=1 TO N
1000 FOR F=1 TO N
1010 IF F(D)<>C THEN GOTO 1030

1020 G(B,F)=E(F(D),F)+G(B,F)
1030 NEXT F
1040 NEXT C
1050 NEXT D
1060 IF V(B,1)=0 THEN GOTO 1100
1070 INPUT "Para ver otro Nivel, pulse RETURN";HH
1080 IF HH=87651! THEN END
1090 GOTO 810
1100 IF R(B,1)=0 THEN PRINT "Nivelaci¢nTerminada"
1110 IF R(B,1)=0 THEN INPUT"PARA MATRIZ DE RESTRICCIONES PULSE RETURN";K
1120 IF R(B,1)=0 AND K<>234 THEN GOTO 1210
1130 IF MM=6736 THEN END
1140 GOTO 1030
1150 REM *** SUBRUTINA DE PERMUTACION*******************
1160 D=D+1
1170 F(D)=F
1180 PRINT "PT n§=";F
1190 M=D
1200 RETURN
1210 REM *** COMIENZO SALIDA MATRIZ DE RESTRICCIONES ***
1220 REM ***********************************************
1230 CLS
1240 PRINT " MATRIZ DE RESTRICCIONES "
1250 PRINT
1260 N=NT
1270 FOR C3=1 TO N
1280 FOR C4=1 TO N
1290 PRINT X(C4,C3);
1300 NEXT C4
1310 PRINT
1320 NEXT C3
1330 INPUT "                    Para Tabla de Alertas, pulse Return";W
1340 CLS
1350 PRINT
1360 PRINT "TABLA DE ALERTAS DE RESTRICCIONES"
1370 PRINT "---------------------------------"
1380 PRINT
1390 N=NT
1400 FOR C3=1 TO N
1410 FOR C4=1 TO N
1420 IF X(C4,C3)>1 THEN GOSUB 1490
1430 NEXT C4
1440 NEXT C3
1450 PRINT
1460 PRINT "FIN DEL PROGRAMA"
1470 END
1480 REM **** SUBRUTINAS DE ALERTA ****
1490 REM Aqu¡o pueden f cilmentea¤adirse las alertas correspondientes a los
1500 REM casos menos radicales citados en el¤ texto. Basta segir el orden
1510 REM num‚rico y la sintaxis que se presenta en las l¡neas siguientes
1520 IF X(C4,C3)=2 THEN PRINT "Entre los puestos de trabajo";C4;" y";C3;"debe existir"
1530 IF X(C4,C3)=2 THEN PRINT "la mayor cercan¡a posible"
1540 IF X(C4,C3)=9 THEN PRINT "Entre los puestos de trabajo";C4;" y";C3;"debe existir"
1550 IF X(C4,C3)=9 THEN PRINT "la mayor distanciaposible"
1560 PRINT
1570 RETURN

